
TP1 : Pgcds d’entiers et Relations de Bézout

N’oubliez pas d’exécuter (valider avec la touche Entrée) les commandes Maple (texte en rouge) 
avant de les utiliser.

L’algorithme d’Euclide récursif et itératif 
On programme les deux versions donnés en cours :
> PgcdRec:=proc(a,b)

if b=0 then a else PgcdRec(b,irem(a,b)) fi

end:

> PgcdIter:=proc(a0,b0)

local r,a,b;

a:=a0;b:=b0;

while b<>0 do

  r:=irem(a,b);a:=b;b:=r

od;

a

end:

On les essaie et on compare à la fonction    igcd  de Maple :
> PgcdRec(105,132),PgcdIter(105,132),igcd(105,132);

, ,3 3 3
> PgcdRec(3443,55),PgcdIter(3443,55),igcd(3443,55);

, ,11 11 11

Voir ce qui se passe dans vos programmes
On veut savoir ce qui se passe pendant l’exécution d’un programme, en particulier lorsqu’on 
cherche des erreurs.
Pour cela un outil : la fonction  trace  
> trace(PgcdRec);PgcdRec(105,132);

PgcdRec
{--> enter PgcdRec, args = 105, 132
{--> enter PgcdRec, args = 132, 105
{--> enter PgcdRec, args = 105, 27
{--> enter PgcdRec, args = 27, 24
{--> enter PgcdRec, args = 24, 3
{--> enter PgcdRec, args = 3, 0

3
<-- exit PgcdRec (now in PgcdRec) = 3}

3
<-- exit PgcdRec (now in PgcdRec) = 3}

3
<-- exit PgcdRec (now in PgcdRec) = 3}

3
<-- exit PgcdRec (now in PgcdRec) = 3}

3
<-- exit PgcdRec (now in PgcdRec) = 3}

3
<-- exit PgcdRec (now at top level) = 3}



3
> trace(PgcdIter);PgcdIter(105,132);

PgcdIter
{--> enter PgcdIter, args = 105, 132

 := a 105

 := b 132

 := r 105

 := a 132

 := b 105

 := r 27

 := a 105

 := b 27

 := r 24

 := a 27

 := b 24

 := r 3

 := a 24

 := b 3

 := r 0

 := a 3

 := b 0

3
<-- exit PgcdIter (now at top level) = 3}

3
Quand vous ne voulez plus de la trace de vos fonctions
> untrace(PgcdRec,PgcdIter);

,PgcdRec PgcdIter

On peut également augmenter la variable  printlevel  qui par défaut vaut 1 mais c’est un outil 
plus difficile à contrôler qui peut-être trop bavard.

Exercice 1 : affichage des restes consécutifs
Modifier les deux fonctions PgcdRec et PgcdIter pour qu’elles affichent les restes calculés au 
fur et à mesure à l’aide de la fonction      print   .
> PgcdRec:=proc(a,b)

print(a);if b=0 then a else PgcdRec(b,irem(a,b)) fi

end:

> PgcdRec(105,132);

105

132

105

27

24

3



3
> PgcdIter:=proc(a0,b0)

local r,a,b;

a:=a0;b:=b0;

while b<>0 do

  print(a);r:=irem(a,b);a:=b;b:=r

od;

a

end:

> PgcdIter(105,132);

105

132

105

27

24

3

Exercice 2 : construction de la suite d’Euclide
Modifier les deux fonctions PgcdRec et PgcdIter de départ pour qu’elles rendent la suite des 
restes calculés et pas seulement le dernier.
> PgcdRec:=proc(a,b)

if b=0 then a,0 else a,PgcdRec(b,irem(a,b)) fi

end:

> PgcdRec(105,132);

, , , , , ,105 132 105 27 24 3 0
> PgcdIter:=proc(a0,b0)

local r,a,b,s;

a:=a0;b:=b0;s:=a,b;

while b<>0 do

  r:=irem(a,b);a:=b;b:=r;s:=s,r

od;

s

end:

> PgcdIter(105,132);

, , , , , ,105 132 105 27 24 3 0

Exercice 3 : relation de Bézout
En s’inspirant des fonctions précédentes écrire des fonctions BezoutIter et BezoutRec qui 
rendent le pgcd et des coefficients de Bézout :
> BezoutIter:=proc(a0,b0)

local r,a,b,q,u0,v0,u1,v1,u2,v2;

a:=a0;b:=b0;u0:=1;v0:=0;u1:=0;v1:=1;

while b<>0 do

  r:=irem(a,b);q:=iquo(a,b);

  u2:=u0-q*u1;v2:=v0-q*v1;

  a:=b;b:=r;u0:=u1;v0:=v1;u1:=u2;v1:=v2

od;

a,u0,v0

end:



> BezoutIter(105,132);

, ,3 -5 4
On vérifie
> -5*105+4*132;

3

> BezoutRec:=proc(a,b,u0,v0,u1,v1)

local q;

if b=0 then a,u0,v0 

else 
q:=iquo(a,b);BezoutRec(b,irem(a,b),u1,v1,u0-q*u1,v0-q*v1) fi

end:


> BezoutRec(105,132,1,0,0,1);

, ,3 -5 4
Si on trouve désagréable et source d’erreurs l’ appel qui initialise u0,v0,u1,v1 on crée une 
procédure non récursive qui fait le premier appel
> BezoutAppRec:=proc(a,b) BezoutRec(a,b,1,0,0,1) end:
> BezoutAppRec(105,132);

, ,3 -5 4

La fonction correspondante de Maple est la fonction  igcdex  .


